



Bergische Universität
Wuppertal

Introduction to the Maple Power Tool intpakX

Walter Krämer

Preprint
BUW-WRSWT 2006/9

Wissenschaftliches Rechnen/
Softwaretechnologie



Impressum

Herausgeber: Prof. Dr. W. Krämer, Dr. W. Hofschuster Wissenschaftliches Rechnen/Softwaretechnologie Fachbereich C (Mathematik und Naturwissenschaften) Bergische Universität Wuppertal Gaußstr. 20 42097 Wuppertal, Germany
--

Internet-Zugriff

Die Berichte sind in elektronischer Form erhältlich über die World Wide Web Seiten

<http://www.math.uni-wuppertal.de/wrswt/literatur.html>

Autoren-Kontaktadresse

Walter Krämer
Bergische Universität Wuppertal
Gaußstr. 20
42097 Wuppertal, Germany
E-mail: kraemer@math.uni-wuppertal.de

Introduction to the Maple Power Tool `intpakX`

Walter Kraemer
University of Wuppertal
42119 Wuppertal, Germany
E-Mail: kraemer@math.uni-wuppertal.de

Abstract

The Maple Power Tool `intpakX` [25] defines Maple types for real intervals and complex disc intervals. On the level of basic operations, `intpakX` includes the four basic arithmetic operators, including extended interval division as an extra function. Furthermore, there are power, square, square root, logarithm and exponential functions, a set of standard functions, union, and intersection. Reimplementations of the Maple construction, conversion, and un-application functions are available. Additionally, there is a range of operators for complex disc arithmetic.

As applications, verified computation of zeros (Interval Newton Method) with the possibility to find all zeros of a function on a specified interval, and range enclosure for real-valued functions of one or two variables are implemented, the latter using either interval evaluation or evaluation via the mean value form and adaptive subdivision of intervals. The user can choose between a non-graphical and a graphical version of the above algorithms displaying the resulting intervals of each iteration step.

The source code (about 2000 lines of Maple-code) of the extension `intpakX` is freely available [24].

Key words: Computer Algebra, Validated Computations, Visualization of Interval Methods, Didactical Tool, Maple Power Tool, `intpakX`

MSC: 65G05, 65G10, 68M15

1 Introduction and Acknowledgement

The Maple Power Tool `intpakX` [9, 25] includes the former packages `intpak` [5, 6], and `intpakX` [7, 8]. It provides the new data type `interval` (long number intervals), the corresponding basic arithmetic operations, basic interval functions, extended interval division, Interval Newton Method, functions to enclose the range of functions, complex basic interval operations (disk arithmetic), the complex exponential function for disc arguments, and a couple of plotting functions to visualize the interval methods. It is particularly useful to illustrate interval algorithms in the field of validating computing. The tool has been used (and will be used) by the author with great success when giving lectures in this field. It is planned to extend the tool by supervising further Bachelor and Master theses.

Please note: `intpakX` assumes that the basic operations provided by Maple are accurate to at least one unit in the last place (ulp) with respect to the actual value of Maple's environment variable `Digits` (number of decimal digits Maple uses to represent software floating-point numbers). If so, it is guaranteed that `intpakX` computes

enclosures for ranges of expressions build from these basic operations. Up to now, a violation of the assumption is not known by the authors of `intpakX`.

A similar statement (1 ulp accuracy) about the accuracy of Maple functions like `exp`, `log`, `sin`, `Bessel`, etc., is probably not correct. Therefore `intpakX` uses several guard digits in its computations. Nevertheless computation of expressions involving such functions can not be guaranteed to enclose the true range! Despite this limitation, we are convinced that `intpakX` is a valuable didactical tool to illustrate interval algorithms/methods. As soon as error bounds for Maple functions will be available, `intpakX` will use the bounds to compute guaranteed enclosures.

The author would like to thank his students/assistants for investing so much time and efforts in this package. Particularly, thanks to Ilse Geuling and Markus Grimmer.

How to install `intpakX` and how to use this Maple Power Tool in your own worksheets is described in the source files, which are publicly available on the WEB (see [24]). From this link you get the most recent version of `intpakX`. The Maplesoft link [25] may present an older version.

2 Real Interval Operations and Functions

A variable x is of type `interval` if x is either an empty list or if $x = [x_1, x_2]$ is a list with two elements. The interval endpoints x_i , $i = 1, 2$ must fulfill one of the following requirements:

- x_i is a real number of type `float`
- x_i is equal to 0
- $x_i \in \{-\text{infinity}, \text{infinity}\}$.

Numbers of type `integer` or of type `fraction` or constants predefined in Maple (to these belong `Pi`, `gamma`, `Catalan`, `FAIL`, `false`, `true`) are not allowed as interval endpoints! However, the `intpakX` function `construct` may be used to create valid intervals in these cases.

Basic Arithmetic Operations: The basic operations implemented are `&+`, `&-`, `&*`, `&/` and `inv`. Their input parameters must be either of type `interval` or of type `num_or_FAIL`. The output is an interval.

A variable is of type `num_or_FAIL` if its value is a number (i. e. the variable is of type `numeric`), `-infinity` or a constant predefined in Maple (see above).

The priorities for the so-called 'inert' operators `&+`, `&-`, `&*`, `&/` are unfortunately set in Maple so that `&+` and `&-` have a higher priority than `&*` and `&/`. Because of this, one has to use lots of brackets in terms!

Basic Interval Functions: The interval functions implemented are `&sqr`, `&sqrt`, `&ln`, `&exp`, `&**`, `&intpower`, `&sin`, `&cos`, `&tan`, `&arcsin`, `&arccos`, `&arctan`, `&sinh`, `&cosh` und `&tanh`. Originally [5] it was assumed that the corresponding mathematical functions in Maple deviate maximally by 0.6 ulp from the exact results. This assumption is, however, confirmed nowhere in the Maple manual. `intpakX` [8] uses several guard digits (concerning reliability refer to the Introduction).

Most names are self-explaining. The operation `&intpower` corresponds to the function x^n , n a natural number. The operator `&**` corresponds to the function x^α , where α can mean an interval, an integer or a number of type `float`.

Auxiliary Functions: The function `construct` generates from a number or a pair of numbers an element of type `interval`. As optional parameter may be entered the string 'rounded'. In this case the interval endpoints are rounded by one ulp from above or from below.

Also contained are the functions `midpoint`, `width`, `&intersect`, `&union` and `is_in`. The function `width` calculates the diameter and `midpoint` an enclosure (confinement) of the centre of an interval.

For converting a term into an interval term resp. into an interval function, there are the commands 'convert/interval' and `inapply`.

Example 1: The data type interval

Interval endpoints of type `integer` are *not* admitted:

```
> x:=[1,2];
                                x := [1, 2]
> type(x,interval);
                                false
> x:=construct(1,2); type(x,interval);
                                x := [1., 2.]
                                true
```

Generate confining interval:

```
> construct(1,rounded);
                                [.999999999, 1.000000001]
> y:=construct(1,infinity,rounded);
                                y := [.999999999, ∞]
> type(y,interval);
                                true
```

Diameter and enclosure of interval centre:

```
> width(x); width(y);
                                1.
                                ∞
> midpoint(x);
                                [1.499999999, 1.500000001]
```

Example 2: Set-theoretic operations

```
> x:=[1.,3.]; y:=[2.,infinity]; z:=[4.,5.];
                                x := [1., 3.]
                                y := [2., ∞]
```

```

                                z := [4., 5.]
> x &union y;
                                [1., ∞]
> x &union z;
                                [1., 3.], [4., 5.]

```

The resulting set consists of two intervals. Thus, `&union` does **not** calculate the interval hull. For this the procedure `&Convex_Hull` is available in `intpakX`.

```

> x &intersect y;
                                [2., 3.]
> x[1];
                                1.
> is_in(x[1],y);
                                false
> is_in(z,y);
                                true

```

Example 3: Arithmetic operations and interval functions

```

> [1.,2.] &+ 3 &* 0; # Wrong priorities of operators
                                [0, 0]
> [1.,2.] &+ (3 &* 0);
                                [.999999999, 2.000000001]
> [-1.,2.] &intpower 3;
                                [-1.000000001, 8.000000001]
> &sqr(&cosh(1)) &- &sqr(&sinh(1));
                                [.9999999919, 1.000000008]

```

Example 4: Transforming expressions into interval functions

```

> f:=inapply(0.5*t,t);
                                f := t → .5‘&*‘t
> f(2);
                                [1.999999999, 3.000000001]
> f:=inapply(sqrt(t),t);
                                f := &sqrt
> f([4.,9.]);
                                [1.999999999, 3.000000001]
> f:=inapply(sin(x)+x,x);
                                f := x → ‘&sin‘(x)‘&+‘x
> f(0);
                                [0, 0]

```

Example 5: Range enclosures

Enclosure of the value domain of $f(x) := x^3 - x^2 - x + 1$ on the interval $[0, 0.5]$ by evaluation with intervals, using the mean value form and taking into consideration the monotony properties of f .

```
> x:='x'; # release variable
```

$$x := x$$

```
> f:=x^3-x^2-x+1;
```

$$f := x^3 - x^2 - x + 1$$

```
> F:=inapply(f,x); # Transformation of f into an interval function
```

$$F := x \rightarrow (x \text{ \&intpower 3 } \text{ \& + } ((-1) \text{ \& * } (x \text{ \&intpower 2 })) \text{ \& + } (((-1) \text{ \& * } x) \text{ \& + } 1))$$

Transformation of the derivative f' of f into an interval function

```
> dF:=inapply(diff(f,x),x);
```

$$dF := x \rightarrow (3 \text{ \& * } (x \text{ \&intpower 2 })) \text{ \& + } ((-2) \text{ \& * } x) \text{ \& + } (-1)$$

Enclosure (confinements) of the value domain of f on the interval $[0,0.5]$

```
> X:=[0,0.5];
```

$$X := [0, .5]$$

```
> mid_X:=midpoint(X);
```

$$mid_X := [.2499999999, .2500000002]$$

```
> r_i:=F(X); # evaluation with intervals
```

$$r_i := [.2499999994, 1.1250000003]$$

```
> r_m:=F(mid_X) &- ( dF(X) &* (X &- mid_X) ); # mean value form
```

$$r_m := [.2031249977, 1.2031250003]$$

The evaluation with intervals of f' on the interval $X = [0, 0.5]$ shows that f is monotonically decreasing in X .

```
> dF([0,0.5]);
```

$$[-2.0000000003, -.2499999985]$$

The exact value domain of f on X is the interval $[0.375, 1]$. A very sharp enclosure of the value domain can be calculated in the following way:

```
> r_e:=construct(F(X[2])[1],F(X[1])[2]);
```

$$r_e := [.3749999993, 1.0000000003]$$

$F(X[2])$ calculates the evaluation with intervals of f at the point $X[2] = 0.5$. $F(X[2])[1]$ gives the lower interval end of $F(X[2])$, thus a **safe** lower bound for the minimum of f on the interval $[0, 0.5]$.

In order to verify if the 'exact' enclosure of the value domain r_e is contained in the intersection of the enclosures r_i and r_m calculated above, e. g. the procedure `is_in` may be used.

```
> is_in(r_e, r_i &intersect r_m);
      true
```

The Procedure Interval_power: In order to enable the evaluation of e. g. the second derivative of $f(x) := \sqrt{x}$

$$f''(x) = -\frac{1}{4}x^{-\frac{3}{2}}$$

with intervals, the procedure `Interval_power` (alias `&**`) allows rational numbers as second parameter.

Calculating $f''(4)$ with a floating point arithmetic

```
> evalf(-1/4*4^(-3/2));
      -.03125000000
```

Evaluation of $f''(4)$ with intervals

```
> df2:=inapply(diff(sqrt(x),x$2),x);
      df2 := x → (-1/4) &* (x &** (3/2))
> df2(4);
      [-.03125000004, -.03124999997]
```

The Procedure is_in: The `intpak`-procedure `is_in` has two input parameters, and verifies if the first parameter is contained in the second (in a set-theoretical sense). As input parameters are allowed variables of type `interval`, numbers of type `numeric` and the values `FAIL`, `infinity` and `-infinity`. The procedure `is_in` gives out a correct result even if one of the numbers is rational or if the length of the entered numbers exceeds the value of the variable `Digits`. Example:

```
> Digits;
      10
> is_in(1.9999999999999999, [2., 2.]);
      false
> is_in(1/3, [0.333333332, 0.3333333333333333]);
      false
```

The Procedure construct: Example

```
> construct(1/3);
      [.3333333333, .3333333334]
```



```

> (1/3) &- 0.3333333333;
                                [0, .1000000001 10-9]
> construct(1.0000000001);
                                [1.0000000000, 1.0000000001]
> 1.0000000001 &- 1;
                                [0, .1000000001 10-8]

```

3 Application I: Ranges With Graphical Output

3.1 Functions in One Variable

Two simple possibilities to confine the range of a function $f : D \subseteq \mathbb{R} \rightarrow \mathbb{R}$ over an interval $[x] \subseteq D$ were presented already in example 5, section 2. Namely the interval-evaluation of f (if it exists) and the mean value form (if the interval-evaluation of f' over $[x]$ exists).

An improved range enclosure is obtained if the interval $[x]$ is partitioned and a range enclosure is calculated over each subinterval. The interval hull of these subrange enclosures is then an enclosure of the range of f over $[x]$. If the partition is continued successively, then the initial range enclosure can be improved successively.

This is realized by the procedure `compute_range`. The procedure demands three input parameters

- a function `f`,
- the start interval `xstart` (may be entered either as interval or as `range`)
- the number of `iterationsteps` to be performed. It is used as an interrupt criterion.

The order of the three parameters mentioned above is compulsory. In addition, the entering of four optional parameters (in any order) is possible

- Remitting a parameter `Nx = n`, `n` an integer greater than or equal to 1, effects the partitioning of the start interval into `n` intervals before all.
- the optional parameter `linear` (`quadratic`) effects the linear (`quadratic`) convergence of the procedure. On entering `linear`, the 'naive' interval evaluation is used for determining the subrange enclosures. If `quadratic` is remitted as parameter, the procedure `compute_combined_range` is used for determining the subrange enclosures, combining interval-evaluation, mean-value form and monotony-test. If none of these two parameters is entered, then in the first three iteration steps interval-evaluation is used for determining the range enclosures, and from step 4 on is used the procedure `compute_combined_range`.

- the optional parameter `adaptive` effects the adaptive partitioning of the current interval list, and therefore generally leads to a calculation-time reduction.
- the optional parameter `colorlist = [color1,color2,...]` determines the colours used for the graphical illustration of each iteration step. `color1`, `color2`, etc. must be colours predefined in Maple, e. g. `blue`, `red`, `green`, `magenta`, `coral`, `brown` etc. This does, however, not influence the illustration of the last iteration step which is always illustrated in yellow.

For reasons of clearness, only the graphical illustration of the last three iteration steps and the function f are given out. The graphical illustration of all iteration steps is, however, stored in the global variable `q`. The variable `q` is a table. If 3 iteration steps were performed, then the entries `q[1]`, `q[2]` and `q[3]` contain the illustrations of each iteration step, however, not the graph of the function. This is stored in the table entry `q[4]`.

Also the calculated range enclosures are stored — in the global variable `r`. It is also a table and `r[i]` contains the range enclosure calculated in the i -th step.

The current partition of the start interval is stored in the global variable `list_of_intervals`. The corresponding subrange enclosures are stored in the variable `list_of_ranges`.

Examples:

Enclosure of the range of the function

```
> f:=x->exp(-x^2)*sin(Pi*x^3);
```

$$f := x \rightarrow e^{(-x^2)} \sin(\pi x^3)$$

over the interval $X := [0.5, 2.]$ using the procedure `compute_range`

```
> X:=[0.5,2.];
```

$$X := [.5, 2.]$$

```
> compute_range(f,X,4);
```

```
initial range confinement =      [-.7788007834, .7788007834]
```

```
range confinement after iteration step 4 =
[-.3233867682, .6103317518]
```

The initial range confinement is $\approx [-0.78, 0.78]$. The range confinement after 4 iteration steps, that is after partitioning into $2^4 = 16$ subintervals, is $\approx [-0.32, 0.61]$. The graphical output can be found in Figure 1.

The same range confinement is obtained already after 1 iteration step if the start interval is partitioned before all into 2^3 intervals and the optional parameter `quadratic` is given:

```
> compute_range(f,X,1,Nx = 2^3,quadratic);
```

```
initial range confinement =      [-.7788007834, .7788007834]
```




Figure 1: refining a range confinement by partitioning the argument domain into subintervals

```
range confinement after partitioning into 8 subintervals =
```

```
[-.3233867682, .6639743998]
```

```
range confinement after iteration step 1 =
```

```
[-.3233867682, .6103317518]
```

Using the parameter `adaptive` can generally reduce the number of subintervals considerably:

```
> compute_range(f, [0.5, 2.], 6, adaptive);
```

```
initial range confinement =      [-.7788007834, .7788007834]
```

```
range confinement after iteration step 6 =
```

```
[-.2834388814, .5563221618]
```

The current partitioning of the start interval is stored in the variable `list_of_intervals`. Therefore, the total number of subintervals can be determined simply at any time. In the example above it is determined in the following way (after 6 iteration steps):

```
> nops(list_of_intervals);
```

Without the parameter `adaptive` the number of subintervals after 6 iteration steps would be $2^6 = 64$.

In order to illustrate only the last iteration step and the function f , the `plots-command display` can be used. The result of the following call-up is found in Figure 2.

```
> display([q[7],q[6]],title='Adaptive partitioning after 6
iterationsteps',titlefont=[TIMES,BOLD,12]);
```




Figure 2: adaptive partitioning

3.2 Functions in Two Variables

The Procedure `compute_range3d` calculates range confinements for real-valued functions in two real variables over a two-dimensional interval $X \times Y$.

Its input parameters are (analogous to `compute_range`, however, **without** the optional parameters `linear/quadratic` and `adaptive`)

- the function f ,
- a real interval, or a domain X ,
- a real interval, or a domain Y ,

- the number of iteration steps to be performed.

The order of these parameters is compulsory. With this procedure, too, can be set a series of optional parameters

- the parameters $N_x = n$ and $N_y = m$ effect a corresponding partitioning of the start interval (axe-parallel right-angle) in the x - y -direction,
- the parameter `colorlist` has the same meaning as in `compute_range`,
- the parameter `cutout = r` determines the strength of the lines in the illustration of the calculated confinements. Here, r should be 0, 1 or a fraction with $0 < r < 1$.

In addition, any options of the `plot3d`-command can be used.

For determining the subrange confinements in `compute_range3d` the 'naive' interval-evaluation is used. In each iteration step the subintervals are partitioned just in one direction. If e. g. two iteration steps are performed, then in the first step is partitioned in x -direction and in the second in y -direction.

Example: Determining a range confinement for the function

```
> f:=(x,y)->exp(-x*y)*sin(Pi*x^2*y^2);
      f := (x, y) → e(-xy) sin(π x2 y2)
```

over th interval $X \times Y = [\pi/8, \pi/2] \times [\pi/8, \pi/2]$.

```
> X:=[evalf(Pi)/8,evalf(Pi)/2]; Y:=X;
      X := [.3926990818, 1.570796327]
      Y := [.3926990818, 1.570796327]

> compute_range3d(f,X,Y,4);

initial range confinement =      [-.8570898115, .8570898115]

range confinement after iteration step 1 =
[-.8570898115, .8570898115]

range confinement after iteration step 2 =
[-.6800891261, .8570898115]

range confinement after iteration step 3 =
[-.6800891261, .8486122905]

range confinement after iteration step 4 =
[-.5093193828, .7559256232]
```

After each iteration step the calculated range confinement is given out. Only the illustration of the last iteration step and the graphical illustration of the function f are given out. Also in this case, the graphical illustrations of the other iteration steps are stored in the global variable \mathbf{q} .

The graphical output of the procedure can, as in any other 3d-graphics in Maple, be edited thereafter with the commands from the graphics-menue. The desired graphic-options can, however, also be called up directly as parameters. E. g. the command

```
> compute_range3d(f,X,Y,3,cutout=9/10,color=yellow,
>               lightmodel=light2,axes=framed,titlefont=[TIMES,BOLD,12],
>               title='range confinement by partitioning into
>               subintervals',);
```

generates the graphics in Figure 3.

4 Application II: Verified Calculation of Zeros With Graphical Illustration

The extension `intpakX` contains a realization of the extended interval-Newton-iteration

$$\begin{cases} [x]^0 & , \text{ real start interval} \\ [x]^{k+1} := N([x]^k) \cap [x]^k & , k = 0, 1, 2, \dots, \end{cases}$$

where

$$N([x]) := m([x]) - \frac{f(m([x]))}{f'([x])}$$

denotes the interval-Newton-operator and $m([x])$ usually the centre of the interval $[x]$.

If $f : D \subset \mathbb{R} \rightarrow \mathbb{R}$ is a continuously differentiable function on D and $[x]^0 \subset D$ a real interval for which the interval-evaluation $f'([x]^0)$ exists, then the interval-Newton-iteration calculates confinements of all zeros of f contained in $[x]^0$. In addition, using this procedure the existence and uniqueness of simple zeros of f in the given start interval can be proved. A more detailed description of this procedure can be found in [11].

The case $0 \in f'([x]^0)$ is permitted! Therefore, for performing this procedure one needs the extended interval division and the subtraction of an extended interval of a real number (extended interval subtraction, see page 14).

4.1 Extended Interval Division and Extended Interval Subtraction

Let \mathbb{IR} be the set of real intervals and

$$\mathbb{IR}^* := \mathbb{IR} \cup \{[-\infty, r] \mid r \in \mathbb{R}\} \cup \{[l, +\infty] \mid l \in \mathbb{R}\} \cup \{[-\infty, +\infty]\}$$




Figure 3: range confinement of a function in two variables

the set of extended intervals.

The definitions of extended interval division and extended interval subtraction used in `intpak` resp. `intpakX` correspond to the definitions used in [20]. The interval operations defined in this way are inclusion-isotonic.

For two real intervals $[x] = [\underline{x}, \bar{x}]$ and $[y] = [\underline{y}, \bar{y}]$, the extended interval division is defined as follows

$$[x]/[y] := \begin{cases} [x] \cdot [1/\underline{y}, 1/\overline{y}], & \text{if } 0 \notin [y] \\ [-\infty, +\infty], & \text{if } 0 \in [x] \text{ and } 0 \in [y] \\ [\underline{x}/\underline{y}, +\infty], & \text{if } \underline{x} < 0 \text{ and } \underline{y} < \overline{y} = 0 \\ [-\infty, \overline{x}/\overline{y}] \cup [\underline{x}/\underline{y}, +\infty], & \text{if } \underline{x} < 0 \text{ and } \underline{y} < 0 < \overline{y} \\ [-\infty, \overline{x}/\overline{y}], & \text{if } \underline{x} < 0 \text{ and } 0 = \underline{y} < \overline{y} \\ [-\infty, \underline{x}/\underline{y}], & \text{if } 0 < \underline{x} \text{ and } \underline{y} < \overline{y} = 0 \\ [-\infty, \underline{x}/\underline{y}] \cup [\underline{x}/\overline{y}, +\infty], & \text{if } 0 < \underline{x} \text{ and } \underline{y} < 0 < \overline{y} \\ [\underline{x}/\overline{y}, +\infty], & \text{if } 0 < \underline{x} \text{ and } 0 = \underline{y} < \overline{y} \\ [], & \text{if } 0 \notin [x] \text{ and } 0 = [y]. \end{cases}$$

As the data type `interval` admits the points `−infinity` and `infinity` as interval endpoints, the extended interval division can be included without difficulty in the interval package. The corresponding command in `intpakX` is called `ext_int_div`. Examples:

```
> ext_int_div([1.,2.], [-1.,1.]);
      [-∞, −.999999999], [.999999999, ∞]
> ext_int_div([-2.,-1.], [0,2.]);
      [-∞, −.499999999]
```

For $r \in \mathbb{R}$ and an interval $[y] \in \mathbb{IR}^* \cup \{[]\}$ the extended interval subtraction is defined by

$$r - [y] := \begin{cases} [r - \overline{y}, r - \underline{y}], & \text{if } [y] = [\underline{y}, \overline{y}] \in \mathbb{IR} \\ [-\infty, +\infty], & \text{if } [y] = [-\infty, +\infty] \\ [r - \overline{y}, +\infty], & \text{if } [y] = [-\infty, \overline{y}] \\ [-\infty, r - \underline{y}], & \text{if } [y] = [\underline{y}, +\infty] \\ [], & \text{if } [y] = []. \end{cases}$$

This is already realized by the subtraction operator `&-` contained in `intpak`. Examples:

```
> 1 &- [1.,infinity];
      [-∞, 0]
> 1 &- [-infinity,1.];
      [0, ∞]
> 1 &- [];
      []
> 1 &- [-infinity,infinity];
      [-∞, ∞]
```


4.2 Extended Interval-Newton-Iteration

The procedure `compute_all_zeros` computes confinements of all zeros of a continuously differentiable function in a given entered start interval using the interval-Newton-iteration.

The input parameters of the procedure `compute_all_zeros` are

- the function `f` whose zeros are supposed to be calculated,
- the start interval `xstart` of the iteration and
- the desired relative diameter `eps` of the zero-confinements to be calculated.

The relative diameter of a real interval is defined as follows

$$d_{rel}([x]) := \begin{cases} \frac{d([x])}{\langle [x] \rangle}, & \text{if } 0 \notin [x] \\ d([x]), & \text{otherwise.} \end{cases}$$

Hereby, $d([x])$ denotes the diameter and $\langle [x] \rangle$ the minimum absolute value of the interval $[x]$.

The order of the parameters mentioned above is compulsory. As optional fourth parameter the desired accuracy, i. e. the value of the system variable `Digits` within the procedure can be entered. The fourth parameter should therefore be a positive integer greater than or equal to 10. If this fourth parameter is missing, then the accuracy is adapted to the relative accuracy needed and the length of the input-parameters, but in any case is greater than or equal to the current value of the variable `Digits`.

Given out is the accuracy used, the computed confinements of the zeros, and for each confinement the information if existence and uniqueness of a zero in the given interval was proved.

If the calculated interval was only a *potential* confinement of a zero, then it may contain either one, various or no zero of f at all.

The computed zero-confinements are stored in a global variable `zeros` and can therefore be used further in any way. `zeros` is a table and the access to an entry of the table is made in the usual way, e. g. using `zeros[2]`.

Further global variables initialized in the procedure are the table `infos` containing the additional informations, the number of calculated zero-confinements `N` and the step counter `iter_counter`.

Example 1: Calculating all zeros of

$$\begin{aligned} > \text{f} := x \rightarrow 2 * \exp(\tan(\cos(x))) - \sin(x) + \cos(2 * x); \\ & \quad f := x \rightarrow 2 e^{\tan(\cos(x))} - \sin(x) + \cos(2x) \end{aligned}$$




Figure 4: illustration of the function $f(x) := 2e^{\tan(\cos(x))} - \sin(x) + \cos(2x)$

in the interval $[0, 8]$.

The function has three simple zeros in the interval $[0, 8]$ (see Figure 4). Two of them can be given exactly, namely $\frac{\pi}{2}$ and $\frac{5\pi}{2}$. An approximation of the third zero can be determined with the command `fsolve`.

Testing if $\pi/2$ and $5\pi/2$ are zeros of f :

```
> f(Pi/2);
                                0
> f(5*Pi/2);
                                0
```

Calculating the third zero with `fsolve`, `Digits=30`:

```
> Digits:=30: zero3:=fsolve(f(x),x=2..2.5); Digits:=10:
      zero3 := 2.26480074200004996505814286126
```

Calculating the zero-confinements, `Digits=20` (fourth input parameter):

```
> compute_all_zeros(f, [0,8.], 10^(-10), 20);
```

```
Digits =    20
```

```
[7.8539816339705772082, 7.8539816339775304044]
```

contains exactly one zero

```
[2.2648007419999768034, 2.2648007420001249007]
```

contains exactly one zero

```
[1.5707963267948966180, 1.5707963267948966204]
```

contains exactly one zero

4.3 Graphical Illustration

For graphical illustration of the interval-Newton-iteration, the procedure `compute_all_zeros_with_plot` is at the user's disposal. It calculates analogously to the procedure `compute_all_zeros` zero-confinements with the interval-Newton-iteration. Additionally, however, each iteration step is illustrated graphically.

Here, too, the value of the variable `Digits` can be entered as an optional fourth parameter. Further, the input of a fifth optional parameter is possible, stating how many iteration steps may be performed maximally. If this fifth parameter is missing, then the maximum number of iteration steps must be entered interactively. The input must end with a colon or a semicolon.

Example: Calculating a zero-confinement of the function

```
> f:=x->exp(sin(x-1))-1;
                                f := x → esin(x-1) - 1
in the interval [0,3.]:
> compute_all_zeros_with_plot(f,[0.,3.],10^(-3));
> 10;
```

The output of the procedure call-up is found on the pages 19 and 20. The slopes of the dotted lines are given by the smallest resp. greatest slope of all tangents to the graph of the function in the current argument domain `xalt`. The lines intersect in the point (expansion point, $f(\text{expansion point})$). The intersection points of these lines with the `x`-axis are important auxiliary quantities for determining the next iteration of the extended interval-Newton-iteration (see page 12).

5 Complex Intervals and Operations (Disc Arithmetics)

Based on the real interval operations, a complex interval arithmetic for disc intervals is defined and implemented in the extension `intpakX`.

A disc interval with centre $z_0 \in \mathbb{C}$ and radius $r > 0$

$$Z = \langle z_0, r \rangle := \{z \in \mathbb{C} \mid |z - z_0| \leq r\}$$

is stored in `intpakX` as a list with three entries: real part of the centre z_0 of Z , imaginary part of z_0 and radius r .

The name of this new data type is `complex_disc`. As components of this new type numbers of type `numeric` are permitted, in particular numbers of type `integer` and of type `fraction`.


Digits = 10

Iteration step 1

$x_{Old} = [0, 3.]$

$x_{New1} = [2.043797652, 3.]$


$x_{New2} = [0, 1.273700327]$




Iteration step 2

$x_{Old} = [0, 1.273700327]$


$x_{New1} = [.8650186701, 1.273700327]$



```
Iteration step    3
xOld=  [.8650186701, 1.273700327]
xNew1=  [.9840864652, 1.014594170]
```



```
Iteration step    4
xOld=  [.9840864652, 1.014594170]
xNew1=  [.9999902275, 1.000010447]
```



[.9999902275, 1.000010447]

contains exactly one zero

```
number of zero confinements:    1
number of iteration steps:      4
```

The procedure `complex_disc_plot` can be used for graphical illustration of a disc interval. It has as input parameter a variable of type `complex_disc`. As further optional parameters the usual illustration options of the Maple-command `plot` can be entered.

5.1 Arithmetic (Disc-)Operations

The basic arithmetic operations for disc intervals are usually defined (see e. g. [1]) as follows. Let $A = \langle a, r_a \rangle$ and $B = \langle b, r_b \rangle$ be two disc intervals. Then

$$\begin{aligned} A + B &:= \langle a + b, r_a + r_b \rangle, \\ A - B &:= \langle a - b, r_a + r_b \rangle, \\ A \cdot B &:= \langle a \cdot b, |a|r_b + |b|r_a + r_a r_b \rangle, \\ 1 / B &:= \left\langle \frac{\bar{b}}{b\bar{b} - r_b^2}, \frac{r_b}{b\bar{b} - r_b^2} \right\rangle, \quad 0 \notin B, \\ A / B &:= A \cdot (1 / B), \quad 0 \notin B, \end{aligned}$$

where $|a| = \sqrt{a_1^2 + a_2^2}$ denotes the absolute value of the complex number $a = a_1 + i a_2$ and $\bar{b} = b_1 - i b_2$ is the complex conjugate of $b = b_1 + i b_2$.

For the operations defined in this way, we have with the notations as above

$$\begin{aligned} A \pm B &= \{a \pm b \mid a \in A, b \in B\}, \\ A \cdot B &\supseteq \{a \cdot b \mid a \in A, b \in B\}, \\ 1 / B &= \{1/b \mid b \in B\}, \\ A / B &\supseteq \{a/b \mid a \in A, b \in B\}, \end{aligned}$$

where the inclusions are distinct from the equalities in general.

The operations for disc intervals defined above are realized in the extension `intpakX` by the operators `&cadd`, `&csub`, `&cmult` and `&cdiv`. For the inversion, there is no extra operator.

Area-Optimal Multiplication and Division: The (so-called **centred**) multiplication of two disc intervals $A = \langle a, r_a \rangle$ and $B = \langle b, r_b \rangle$ defined above delivers, for given centre $a \cdot b$, an optimal confinement of the resulting point complex $\{\alpha \cdot \beta \mid \alpha \in A, \beta \in B\}$. This confinement, however, is not area-optimal.

Determining an area-optimal confinement of the resulting set under multiplication of two disc intervals is more tedious and leads to solving an equation of third degree (see [15]).

For $A = \langle a, r_a \rangle$ and $B = \langle b, r_b \rangle$, the **area-optimal** multiplication is defined as

$$A \cdot_{opt} B := \langle ab(1 + x_0), (3|ab|^2 x_0^2 + 2(|ab|^2 + |r_a b|^2 + |r_b a|^2) x_0 + |r_a b|^2 + |r_b a|^2 + (r_a r_b)^2)^{\frac{1}{2}} \rangle$$

where x_0 is the non-negative zero of the polynomial

$$P(x) = 2|ab|^2 x^3 + (|ab|^2 + |r_a b|^2 + |r_b a|^2) x^2 - r_a^2 r_b^2$$

if $\text{grad}(P) \geq 2$ (otherwise, we set $x_0 = 0$).

The area-optimal division of two disc intervals is then defined as

$$A /_{opt} B := A \cdot_{opt} (1 / B).$$

The package `intpakX` contains a realization of the area-optimal multiplication (`&cmult_opt`) and the area-optimal division (`&cdiv_opt`) of two disc intervals.

General Procedure when Implementing the Basic Operations: Let A and B be two disc intervals, which can be displayed on the calculator exactly and let $*$ $\in \{+, -, \cdot, /\}$. In order to obtain a safe confinement C on the machine of the exact result complex $A * B$, during the implementation was proceeded as follows:

1. Calculate a real machine interval cx confining the real part of the centre of the result interval, and a real machine interval cy confining the imaginary part.
2. Calculate the radius r of the resulting circle as:

$$\begin{aligned} r1 &:= \text{sup}(\text{formula for the radius evaluated with intervals}) \\ r2 &:= \Delta(r1 + d(cx)) \\ r &:= \Delta(r2 + d(cy)) \end{aligned}$$

where Δ denotes the rounding from above and $d(cx)$, $d(cy)$ the diameter of cx resp. cy .

3. Set $C = \langle m(cx) + i \cdot m(cy), r \rangle$. $m(cx)$ and $m(cy)$ denote the centre of cx resp. cy .

In order to determine the centre of an interval, the procedure `mid` is used. In contrast to the `intpak`-procedure `midpoint`, it does not calculate a confinement of the centre of an interval, but a number (approximation of the centre of the interval) lying certainly within the entered interval.

A Numerical Example: For $A = \langle 1, 1 \rangle$, $B = \langle -1 + i, 1 \rangle$ we have

$$\begin{aligned} A + B &= \langle i, 2 \rangle \\ A - B &= \langle 2 - i, 2 \rangle \\ 1 / B &= \langle -1 - i, 1 \rangle \end{aligned}$$

Calculation with Maple

```
> A:=[1,0,1]: B:=[-1,1,1]:
> A &cadd B;
                                [0, 1.000000000, 2.000000005]
> A &csub B;
                                [2.000000000, -1.000000000, 2.000000007]
> 1 &cdiv B;
                                [-1.000000001, -1.000000001, 1.000000051]
```

Using centred multiplication (see p. 21) one obtains

$$A \cdot B = \langle -1 + i, 2 + \sqrt{2} \rangle \approx \langle -1 + i, 3.414213562 \rangle$$

$$A / B = \langle -1 - i, 2 + \sqrt{2} \rangle$$

```
> A &cmult B;
                                [-1.000000000, 1.000000000, 3.414213579]
> A &cdiv B;
                                [-1.000000001, -1.000000001, 3.414213685]
```

and using area-optimal multiplication leads to

```
> A &cmult_opt B;
                                [-1.390388204, 1.390388204, 2.969562256]
> A &cdiv_opt B;
                                [-1.390388219, -1.390388219, 2.969562345]
```

In Figure 5 on p. 24 are displayed simultaneously the resulting point set of the product of A and B , the centred confinement and the area-optimal confinement of the set.

6 Application III: Range of Complex Polynomials

A first possible application of the in `intpakX` defined disc arithmetic is the determination of safe confinements for the range of a polynomial with complex coefficients over a disc interval.

For this, there are three procedures

1. `horner_eval_cent` (Horner-scheme using centred multiplication `&cmult`),
2. `horner_eval_opt` (Horner-scheme using area-optimal multiplication `&cmult_opt`),
3. `centred_form_eval` (centred form for complex polynomials).




Figure 5: centred and area-optimal confinement of $\langle 1, 1 \rangle \cdot \langle -1 + i, 1 \rangle$

The procedures `horner_eval_opt` and `centred_form_eval` generally give certainly better confinements than the procedure `horner_eval_cent`. They have, however, a substantially higher demand for time and memory.

The first input parameter of each procedure is a (complex) polynomial in the variable z . The denomination of this variable is compulsory! As second parameter must be entered a number or variable of type `complex_disc`.

Example 1: Confining the range of

$$\begin{aligned}
 p(z) := & (0.15 - 0.1i) + (0.15 - 0.12i)z + (-0.2 - 0.2i)z^2 \\
 & + (0.1 + 0.3i)z^3 + (0.1 - 0.2i)z^4 + (0.1 - 0.2i)z^5 \\
 & + (0.2 - 0.2i)z^6 + (0.1 - 0.2i)z^7 + (0.2 - 0.1i)z^8 \\
 & + (0.1 - 0.1i)z^9
 \end{aligned}$$

over the interval $Z = \langle -0.1 + 0.2i, 0.9 \rangle$.

```

> p_H:=horner_eval_cent(p,Z);
    p_H := [.1590115281, -.04050517670, 3.058832329]
> p_Hopt:=horner_eval_opt(p,Z);
    p_Hopt := [.2219721872, .2917174855, 2.243412729]
> p_C:=centred_form_eval(p,Z);
    p_C := [.1590115281, -.04050517670, 1.717944237]

```

In this example the procedure `centred_form_eval` gives the best confinement. The graphical illustration of the range of p over Z and the calculated confinements can be found in Figure 6.




Figure 6: confinement of the range of a complex polynomial

Generating the graphics:

For illustrating the range of p over Z the command `complexplot` from the `plots`-package can be used. Each command should be ended by a colon (otherwise very much, generally unnecessary, information is given out).

```
> c1:=complexplot(subs(z=Z[1]+I*Z[2]+Z[3]*(cos(t)+I*sin(t)),p),
                  t=0..2*Pi,color=black,thickness=3,numpoints=200):
```

For graphical illustration of the computed confinements the command `complex_disc_plot` was used.

```
> c2:=complex_disc_plot(p_H,color=black,thickness=2,linestyle=4):
> c3:=complex_disc_plot(p_Hopt,color=black,thickness=2,linestyle=3):
> c4:=complex_disc_plot(p_C,color=black,thickness=2):
```

The labelling of the graphic was generated with a series of `plot`- and `textplot`-commands. `textplot` is a command from the `plots`-Package.

```
> s1:=plot([[2.1,2.3],[2.6,2.6]],color=black,thickness=2):
> t1:=textplot([3,2.8,pH],font=[TIMES,BOLDITALIC,12]):
> s2:=plot([[1.5,1],[3.2,1.8]],color=black,thickness=2):
> t2:=textplot([3.7,2,pC],font=[TIMES,BOLDITALIC,12]):
```


A detailed discussion of images of disc intervals under the exponential function is found in [4].

The realization of the exponential function for disc intervals `cexp` from the extension `intpakX` has as input parameter a variable of type `complex_disc` or a complex number and gives a safe confinement of the resulting point complex.

Example: Confinement of the image of $Z = \langle 0, \pi + 1 \rangle$ under the exponential function

```
> Cexp:=cexp([0,0,evalf(Pi+1)]);
      Cexp := [1.000000000, 0, 61.90292461]
```

Graphical illustration of the computed disc interval confinement:

```
> c1:=complex_disc_plot(Cexp,color=black,thickness=3,numpoints=400):
Image of the boundary of  $\langle 0, \pi + 1 \rangle$  under the exponential function (Attention:
plots must be loaded beforehand with with!):
```

```
> c2:=complexplot(exp(polar(Pi+1,phi)),phi=0..2*Pi,
                  color=black,thickness=3,numpoints=400):
```

For illustrating inner points of the range, the radius r is varied from 0 to $\pi + 1$ for fixed angle. Example:

```
> c2:=complexplot(exp(polar(r,0.5)),r=0..Pi+1,
                  color=black,thickness=3,numpoints=400):
```

The different graphic commands are bundled afterwards with `display`. On display should be entered as additional option `scaling=constrained`. In order to display a cut out piece, the `plot-option view` was used.

The graphical illustration in Figure 7 shows that the centred confinement of the image of $\langle 0, \pi + 1 \rangle$ under the exponential function is the optimal confinement for prescribed result-centre $\exp(0) = 1$. It is far from area-optimal. Figure 8 shows a cut out piece around the origin.

The exponential function is $2\pi i$ periodic, and as the diameter of the interval $Z = \langle 0, \pi + 1 \rangle$ is greater than 2π , there is an area in the image Z of the exponential function, in which every point has exactly two preimages. This area can be recognized by its double hatching. Besides, in the illustration of the cut out piece can be observed the drop-like image-free area around the origin.

8 Conclusion and Outlook

Using verification algorithms it is possible, if occasion arises, to prove with the computer automatically the existence and uniqueness of a solution for a given problem, and also to compute a (narrow) confinement of the exact solution. The results obtained in this way have the same mathematical quality as results obtained e. g. by using computer




Figure 7: image of the disc interval $\langle 0, \pi + 1 \rangle$ under the exponential function and centred disc interval confinement of $\exp(\langle 0, \pi + 1 \rangle)$

algebra systems, i. e. by automatic formula manipulations. Thereby it turns out to be a great advantage that verification algorithms can handle safely numerical input-data with errors within tolerance. In such cases infinitely many problems are solved simultaneously. For an entire family of problems it is e. g. proved that each has a unique solution.

Whenever computer algebra packages make use of numerical routines (e. g. when computing determined integrals), a verification algorithm should be used if possible (e.g. [2]). The results obtained are then mathematically safe (a property usually expected when working with a CA-system). Pretended solutions resp. far off approximations are then excluded.

Also by using interval methods, the graphic abilities of computer algebra systems can be improved resp. made secure. That this is necessary, is impressively shown by [16, 14]. See also [3, 8, 17, 21].

Computer algebra and verification numerics complete each other ideally. Thus the computer becomes for the mathematician, but also for the engineer, a safe mathematical tool. Especially in view of processors which are becoming faster and faster and more powerful, the symbiosis of symbolic calculus and safe numerical routines [1, 18, 11] should be pushed forward massively.

`intpakX` is also intended to be a valuable didactical tool in teaching selfverifying (numerical) methods. This tool allows learning by experience. Because the source code is publicly available, it should be easy to extend the functionality of the package by students/users. The multiple precision interval arithmetic [10] realized in `intpakX` allows the implementation of algorithms as described in [23, 22].




Figure 8: illustration of a cut out piece around the origin

References

- [1] Götz Alefeld, Jürgen Herzberger: *Introduction to Interval Computations*, New York: Academic Press, 1983.
- [2] Thomas Bauknecht: *Verifizierte numerische Quadratur in Maple*, Diplomarbeit (Betreuer: W. Krämer), Univ. Karlsruhe, 1997.
- [3] Ulrich Bolz: *Verifizierte graphische Darstellung reeller Funktionen*, Diplomarbeit (Betreuer R. Lohner), Univ. Karlsruhe, 1996.
- [4] Norbert C. Börsken: *Komplexe Kreis-Standardfunktionen*, Diplomarbeit, Univ. Freiburg, 1978.
- [5] Amanda E. Connell, Robert M. Corless: *An Experimental Interval Arithmetic Package in Maple*, Tex-Document distributed with the Maple Share Library, 1993.
- [6] George F. Corliss: *INTPAK for Interval Arithmetic in Maple: Introduction and Applications*, J. Symbolic Computation 11, 1994.
- [7] Ilse Geulig: *Computeralgebra und Verifikationsalgorithmen*, Diplomarbeit (Betreuer: W. Krämer), Univ. Karlsruhe, 1998.
- [8] Ilse Geulig und Walter Krämer: *Intervallrechnung in Maple - Die Erweiterung intpakX zum Paket intpak der Share-Library*, Preprint 99/2 des Instituts für Wissenschaftliches Rechnen und Mathematische Modellbildung (IWRMM), Universität Karlsruhe, 1999.

- [9] Markus Grimmer.: *Interval Arithmetic in Maple with intpakX*. PAMM - Proceedings in Applied Mathematics and Mechanics, Vol. 2, Nr. 1, p. 442-443, Wiley-InterScience, 2003.
- [10] Markus Grimmer, Knut Petras, Natalie Revol.: *Multiple Precision Interval Packages: Comparing Different Approaches* Published in: Numerical Software with Result Verification, Lecture Notes in Computer Science, Volume 2991/2004, Springer-Verlag, Heidelberg, pp. 64 - 90, 2004.
- [11] Rolf Hammer, Matthias Hocks, U. Kulisch, D. Ratz: *Numerical Toolbox for Verified Computing I*, Berlin, Heidelberg: Springer-Verlag, 1993.
- [12] Michael Kofler: *Maple: An Introduction and Reference*, Addison-Wesley, 1997.
- [13] Walter Krämer: *Computeralgebra und Verifikationsalgorithmen I und II*, Vorlesungen im WS 96/97 bzw. SS 97, Univ. Karlsruhe.
- [14] Walter Krämer: *Pitfalls in Maple*. Preprint BUW-WRSWT 2006/7, Universität Wuppertal, to appear 2006.
- [15] Norbert Krier: *Komplexe Kreisarithmetik*, Dissertation, Univ. Karlsruhe, 1973.
- [16] Michael Lerch, German Tischler, Jürgen Wolff Von Gudenberg, Werner Hofschuster, Walter Krämer: *FILIB++*, a fast interval library supporting containment computations. ACM Transactions on Mathematical Software (TOMS), Volume 32 Issue 2, 2006.
- [17] Rudolf Lohner: *Private communication about questionable Maple results*, 1998.
- [18] Arnold Neumaier: *Interval Methods for Systems of Equations*, Cambridge: Cambridge University Press, 1990.
- [19] H. Ratschek, J. Rokne: *Computer Methods for the Range of Functions*, Chichester, West Sussex, England: Ellis Horwood Limited, 1984.
- [20] Dietmar Ratz: *Inclusion Isotone Extended Interval Arithmetic*, Bericht 5/96, Institut für Angewandte Mathematik, Univ. Karlsruhe, 1996.
- [21] Evgenija Popova: *Web-accessible tools for interval linear systems*. PAMM 5, issue 1, pp. 713–714, 2005.
- [22] Rogat, Axel: *Schnelle hoch genaue Einschließung von Werten arithmetischer Ausdrücke mit beliebiger vorgegebener Genauigkeit*. Dissertation, Universität Wuppertal, 2002.
- [23] A. Steins: *Verifizierte Formelauswertung in Computer-Algebra-Systemen*, Dissertation, Universität Wuppertal, 1996.
- [24] `intpakX` link at the University of Wuppertal:
<http://www.math.uni-wuppertal.de/~xsc/software/intpakX/>

[25] Maplesoft: [http://www.maplesoft.com/applications/
app_center_browse.aspx?CID=13&SCID=155](http://www.maplesoft.com/applications/app_center_browse.aspx?CID=13&SCID=155)